INTRODUCING APL TO TEACHERS

IBM Philadelphia Scientific Center
Data Processing Division
TECH. REPORT NO. 320-3014 JULY 1972

KENNETH E. IVERSON

IBM PHILADELPHIA SCIENTIFIC CENTER TECHNICAL REPORT NO. 320-3014 JULY 1972

INTRODUCING APL TO TEACHERS

K. E. IVERSON

PHILADELPHIA SCIENTIFIC CENTER
IBM CORPORATION
3401 MARKET STREET, PHILADELPHIA, PENNSYLVANIA, 19104

PREFACE

In introducing the use of a computer to teachers it is desirable to start as soon as possible with material which they can see is relevant to their topic and their students, and to avoid digressions concerning the computer and computer language. This paper presents such an introduction to APL for teachers of high school mathematics. Much of this material should also be suitable for teachers of other topics at other levels, although they would also benefit from auxiliary material specifically addressed to the topic of interest.

The teachers are expected to spend most of their time at an APL terminal, with ideally two and at most four per terminal. Page 1 gives general information and each succeeding page presents instructions and work for one session, at the end of which the instructor answers questions and gives any comments necessary to introduce the next session.

Each session should be fifteen to twenty minutes in length, although an interruption for questions and discussion could be followed by a decision to continue work on the same page. Teachers should not look ahead since a later page may explain something which might better be learned by the experimentation suggested. The instructor and assistants should circulate among the groups offering any necessary assistance and collecting points for discussion at the next interruption. Necessary assistance should not include information which the teachers can easily be led to acquire for themselves by experimentation.

The terminals should be so located that teachers need not leave them for the inter-session discussion, but terminal use must not be allowed to interrupt the discussion. An overhead projector and transparencies made from the pages are very useful aids in the discussion. A transparency maker can also be very helpful, allowing the immediate display of examples produced by the teachers or instructor.

An instructor with limited experience with APL should be able to give specific answers to the questions which arise. More helpful general answers can only be expected from an instructor well-versed in APL and its philosophy. The instructor may find "Algebra as a Language" (which appears as Appendix A in Reference 2) helpful in this regard and may also wish to recommend it to teachers for later reading.

If the pages are used in sequence each bit of notation needed is introduced before it is used. However, the dependence between the pages is minimal and the order of presentation can be varied widely, particularly if the instructor is prepared to fill any gaps on demand.

TABLE OF CONTENTS

Preface	i
Introduction	1
Experimentation	2
Systematic Experimentation	3
Multiplication and Other Function Tables	4
Graphs and Bar Charts	5
Indexing and Characters	6
Exploring Functions of One Argument	7
Defining New Functions	8
Inverse Functions	9
Summation and Other Functions Over a List	10
Factoring	11
Linear Expressions	12
Linear Equations	13
Tables and Graphs of Linear Functions	14
Polynomials	15
Generalizing a Function by Use of Patterns	16
The Positive Integers	17
Summation of Series	18
Power Series	19
Differencing a Function	20
Combinations and Binomial Coefficients	21
Iteration	22
References to Other Topics	23
References	23
Overlay for Tables	24
Summary of Notation	25

INTRODUCTION

1. PURPOSE

To show the use of the APL terminal in teaching high school mathematics. The topics treated are chosen for their value as isolated examples and are not to be construed as a proposal for a course.

2. APPROACH

Let you use the terminal.

3. THE TERMINAL

An ordinary typewriter but for two characteristics:

A. A typing element with mathematical symbols.

B. A device to encode each keystroke in audible tones for transmission via telephone to a computer which responds with a similarly encoded transmission.

4. PROCEDURE

Work from one page for 15 to 20 minutes, then stop for discussion before proceeding to the next page. Do not look ahead.

5. GENERAL ADVICE

A. Every statement must be concluded by a carriage return.

B. Don't hesitate to try <u>anything</u>; no harm can result.

C. If you do not understand the result produced by an expression, try a related expression which might yield further clues.

D. Do not spend too much time on any one difficulty, but raise it as a question in the discussion period between pages.

EXPERIMENTATION

A. Simple expressions:

```
      3+4                    Carriage Return
7

      3×4.7                  Carriage Return
14.1
```

B. Determine the meanings of the following eight functions (whose locations on the keyboard are identified by arrowheads):

− ÷ * ⌈ ⌊ ≤ = ≠



For example, enter

```
      3-4
¯1
```

to verify that - represents subtraction, and

```
      3÷4
0.75
```

to verify that ÷ represents division.

SYSTEMATIC EXPERIMENTATION

A. On single quantities:

```
      2|1
1
      2|2        Vary one argument systematically.
0
      2|3
1
```

B. On lists of numbers:

```
      3|1 2 3 4 5 6 7
1 2 0 1 2 0 1
```

C. Use names for convenience:

```
      X←5

      X*2
25
      S←1 2 3 4 5 6 7

      3|S
1 2 0 1 2 0 1

      S*3
1 8 27 64 125 216 343

      S×S
1 4 9 16 25 36 49
```

D. Explore the functions of Page 2 Part B for negative numbers. For example:

```
      T←S-4

      T*2
9 4 1 0 1 4 9
```

MULTIPLICATION AND OTHER FUNCTION TABLES

A. Expressions for tables:

```
      S←1 2 3 4 5 6 7

      S∘.×S
1  2  3  4  5  6  7
2  4  6  8 10 12 14
3  6  9 12 15 18 21
4  8 12 16 20 24 28
5 10 15 20 25 30 35
6 12 18 24 30 36 42
7 14 21 28 35 42 49
```

```
      S∘.+S
2  3  4  5  6  7  8
3  4  5  6  7  8  9
4  5  6  7  8  9 10
5  6  7  8  9 10 11
6  7  8  9 10 11 12
7  8  9 10 11 12 13
8  9 10 11 12 13 14
```

```
      B←2 3

      B∘.×S
2  4  6  8 10 12 14
3  6  9 12 15 18 21
```

```
      2 3∘.+S
3  4  5  6  7  8  9
4  5  6  7  8  9 10
```

B. Produce function tables for ⌈ ⌊ < = and |.

To aid in reading the tables you may wish to enter (by hand) the first argument in a column at the left of the table and the second in a row along the top, or overlay the table with a transparency made from page 24.

C. Examine the tables for patterns and try to see why each function generates the particular pattern.

D. Repeat Parts A-C with the vector *T*←*S*-4 replacing *S*.

GRAPHS AND BAR CHARTS

A. Graph of a parabola:

```
      X←1 2 3 4 5 6 7

      V←(X-3)×(X-5)

      V
8  3  0  ¯1  0  3  8

      R←8 7 6 5 4 3 2 1 0 ¯1

      R∘.=V
 1 0 0 0 0 0 1
 0 0 0 0 0 0 0
 0 0 0 0 0 0 0
 0 0 0 0 0 0 0
 0 0 0 0 0 0 0
 0 1 0 0 0 1 0
 0 0 0 0 0 0 0
 0 0 0 0 0 0 0
 0 0 1 0 1 0 0
 0 0 0 1 0 0 0
```

B. Bar chart:

```
      R∘.≤V
 1 0 0 0 0 0 1
 1 0 0 0 0 0 1
 1 0 0 0 0 0 1
 1 0 0 0 0 0 1
 1 0 0 0 0 0 1
 1 1 0 0 0 1 1
 1 1 0 0 0 1 1
 1 1 0 0 0 1 1
 1 1 1 0 1 1 1
 1 1 1 1 1 1 1
```

C. Graph other functions of one argument.

INDEXING AND CHARACTERS

A. Indexing:

```
      X←2 3 5 7 11
      X[4]
7
      X[1 2 3]
2 3 5

      X[5 4 3 2 1]
11 7 5 3 2

      X[4 1 3]
7 2 5
```

B. Characters:

```
      W←'DOG'
      W[3]
G
```

(If your computer gives no response to your entries you may be "in an open quote". Try entering a single quote to escape.)

```
      W[3 2 1]
GOD

      'ABCDEFGHI '[8 5 1 4 10 3 8 9 5 6]
HEAD CHIEF

      ' *'[2 1 2 2 1 2 2 1 2]
* ** ** *
```

C. Plotting:

Enter the following:

```
      X←1 2 3 4 5 6 7

      V←(X-3)×(X-5)

      R←8 7 6 5 4 3 2 1 0 ¯1

      R∘.=V

      ' *'[1+R∘.=V]

      ' *'[1+2≥X∘.-X]
```

EXPLORING FUNCTIONS OF ONE ARGUMENT

A. Negation:

```
      X←3

      -X
¯3

      P←1 2 3 4 5 6 7

      Q←P-4

      R←P÷2

      -P
¯1  ¯2  ¯3  ¯4  ¯5  ¯6  ¯7

      -Q
3  2  1  0  ¯1  ¯2  ¯3

      -R
¯5  ¯1  ¯1.5  ¯2  ¯2.5  ¯3  ¯3.5
```

B. Explore the following functions of one argument:

```
      ÷ | ⌊ ⌈ ⋆
```

[Note that each of these symbols denotes either a function of two arguments (as in *X*÷*Y*) or of one argument (as in ÷*Y*) just as the symbol - denotes either <u>subtraction</u> (as in *X*-*Y*) or <u>negation</u> (as in -*Y*) in conventional notation.]

C. Enter the following expressions:

```
      T←¯3 ¯2 ¯1 0 1 2 3

      T∘.×T

      ×T∘.×T

            '- +'[2+×T∘.×T]
```

Use these results (and any other experiments you wish to try) to determine the meaning of the function × when applied to one argument.

DEFINING NEW FUNCTIONS

A. A parabola with zeros at 3 and 5:

```
      X←7

      (X-3)×(X-5)
8

      ∇Z←F X
[1]   Z←(X-3)×(X-5)∇

      F 7
8
      2×F 7
16

      F F 7
15

      F 1 2 3 4 5 6 7
8  3  0  ¯1  0  3  8
```

(If you wish to change a function *F* after having defined it, type:
)*ERASE F*
Then begin your new definition of *F*.)

B. A test for divisibility by 7:

```
      ∇Z←D X
[1]   Z←0=7|X∇

      D 868
1

      D 6 7 8 9 10 11 12 13 14
0  1  0  0  0  0  0  0  1
```

C. A plotting function.

Enter the following:

```
      ∇Z←PLOT T
[1]   Z←' *'[1+T]∇

      R←8 7 6 5 4 3 2 1 0 ¯1

      PLOT R∘.=F 1 2 3 4 5 6 7
```

INVERSE FUNCTIONS

A. Two simple inverse functions:

```
      ∇Z←F1 X
[1]   Z←3+X∇

      ∇Z←F2 X
[1]   Z←(-3)+X∇
```

These functions are called __inverses__ because one will undo the work of the other:

```
      F1 5
8

      F2 8
5

      F2 1 2 3 4
¯2 ¯1 0 1

      F1 F2 1 2 3 4
1 2 3 4
```

B. Define and test a function *G*2 which is inverse to *G*1:

```
      ∇Z←G1 X
[1]   Z←3×X∇
```

C. Define and test a "Fahrenheit to Centigrade" function *T*2 which is inverse to *T*1:

```
      ∇Z←T1 X
[1]   Z←32+(1.8×X)∇
```

Note: To change any function *G* which has been defined, type:

```
)ERASE G
```

Then proceed with the new definition of *G*.

SUMMATION AND OTHER FUNCTIONS OVER A LIST

A. Summation:

```
      +/3 1 7 4 2 1
18

      3+1+7+4+2+1
18

      X←3 1 7 4 2 1

      +/X
18

      +/(X×X)
80
```

B. Other functions:

```
      ×/3 1 7 4 2 1
168

      3×1×7×4×2×1
168

      ×/X
168

      ⌈/X
7

      X=⌈/X
0 0 1 0 0 0

      ⌊/X
1

      X=⌊/X
0 1 0 0 0 1

      +/X=⌊/X
2
```

C. Experiment with various functions over various lists.

FACTORING

A. Products of powers of primes.

Enter the following expressions:

```
      P←2 3 5 7 11
      E←2 0 2 1 0

      P⋆E
4  1  25  7  1

      M←×/P⋆E
      M
700

      A←1 2 0 1 0

      P⋆A
2  9  1  7  1

      N←×/P⋆A
      N
126
```

B. Greatest common divisor.

Part A shows that E represents the prime factorization of M and that A represents the prime factorization of N. Use some experiments (or a simple deduction) to convince yourself that the expression `×/P⋆(E⌊A)` yields the greatest common divisor of M and N.

C. What is the relation between the results `×/P⋆E` and `×/P⋆A` and `×/P⋆(E⌈A)`. Perform some experiments to confirm your conclusion.

D. What is the relation between `×/P⋆E` and `×/P⋆A` and `×/P⋆(E+A)`.

E. Comment on the following: The number of distinct divisors of the integer `×/P⋆E` is equal to `×/E+1`.

LINEAR EXPRESSIONS

A. Linear expressions.

If *X*←3 5, then the following linear expressions can be evaluated by simply entering them:

```
      (2×X[1])+(4×X[2])
26
      (7×X[1])+(3×X[2])
36
```

B. Coefficient vectors.

The coefficients in each of the expressions can be collected in vectors as follows:

```
      C←2 4
      D←7 3
      C×X
6  20
      +/C×X
26
      +/D×X
36
```

C. Coefficient matrix.

The matrix of coefficients can be formed by <u>reshaping</u> the list of coefficients 2 4 7 3 as follows:

```
      M←2 2⍴2 4 7 3
      M
2  4
7  3
      (M[1;1]×X[1])+(M[1;2]×X[2])
26
      M[1;]
2  4
      +/M[1;]×X
26
      +/M[2;]×X
36
```

D. The matrix product.

The function +.× is the <u>matrix product</u>:

```
      M+.×X
26  36
      M+.×5 ¯2
2  29
```

LINEAR EQUATIONS

A. Solving the simultaneous equations (M+.×X)=B:

```
      M←2 2ρ2 4 7 3

      M
2  4
7  3

      B←2 29

      B⌹M
5  ¯2

      M+.×5  ¯2
2  29

      X←16 23⌹M

      X
2  3

      M+.×X
16  23
```

B. Higher order systems of equations:

```
      M←4 4ρ¯6 6 0 2 ¯4 ¯8 4 4 9 ¯1 1 7 ¯8 ¯7 2 4

      M
 ¯6  6  0  2
 ¯4 ¯8  4  4
  9 ¯1  1  7
 ¯8 ¯7  2  4

      B←20 16 69 1
      X←B⌹M

      X
2  3  5  7

      M+.×X
20  16  69  1
```

TABLES AND GRAPHS OF LINEAR FUNCTIONS

A. Enter the following expressions to obtain a table of the values of the linear function (2×*Y*)+(*X*-12):

```
Y←8 7 6 5 4 3 2 1 0
X←0 1 2 3 4 5 6 7 8

(2×Y)∘.+(X-12)
```

B. Evaluate (2×*Y*)+(*X*-12) for various pairs of single values of *X* and *Y* and compare with the result shown in the table produced in Part A.

C. Enter the following expressions and interpret the significance of the results with respect to the line determined by the linear equation ((2×*Y*)+(*X*-12))=0:

```
E1←((2×Y)∘.+(X-12))=0
E1
L1←((2×Y)∘.+(X-12))<0
L1
U1←((2×Y)∘.+(X-12))>0
U1
```

D. Repeat Parts A and C for the linear function *Y*+(¯1×*X*), naming the results *E*2, *L*2, and *U*2.

E. Enter the following expressions (a table for the functions <u>or</u> (∨) and <u>and</u> (∧) appears at the right):

```
E1∨E2
E1∧E2
L1∧U2
L1∨U2
L2∨U2
L2∨U2∨E2
```

		∧	∨
0	0	0	0
0	1	0	1
1	0	0	1
1	1	1	1

F. State in words an interpretation of each of the results obtained in Part E.

G. Apply the *PLOT* function of page 8 to each of the results of Part E.

POLYNOMIALS

A. Evaluating a polynomial:

```
      C←3 1 4 2              Vector of coefficients

      X←2                    Argument value

      X*0 1 2 3              Powers of argument
1  2  4  8

      C×X*0 1 2 3            Terms of the polynomial
3  2  16  16

      +/C×X*0 1 2 3          Sum of terms
37
```

B. The product of two polynomials:

```
      D←2 0 2 ¯3 1
      D×X*0 1 2 3 4
2  0  8  ¯24  16

      +/D×X*0 1 2 3 4
2
      (+/C×X*0 1 2 3)×(+/D×X*0 1 2 3 4)
74
```

C. The coefficients of the product polynomial:

```
      C∘.×D
 6   0   6   ¯9   3                          The product of all
 2   0   2   ¯3   1                          pairs from the two
 8   0   8  ¯12   4                          vectors of
 4   0   4   ¯6   2                          coefficients
                     C∘.×D
                  6   0   6   ¯9   3         summed as indicated by
                  2   0   2   ¯3   1         the arrows
                  8   0   8  ¯12   4
                  4   0   4   ¯6   2

6   2  14  ¯3   8  ¯7  ¯2   2               produce the coefficients
                                           of the product polynomial.

      E←6 2 14 ¯3 8 ¯7 ¯2 2
      +/E×X*0 1 2 3 4 5 6 7
74
      (+/C×X*0 1 2 3)×(+/D×X*0 1 2 3 4)
74
```

GENERALIZING A FUNCTION BY USE OF PATTERNS

A. The pattern shown by a function:

```
      3×1 2 3 4
3  6  9  12
```

Add 3 to get next entry to right.
Subtract 3 to get next entry to left.

```
      2*2 3 4 5
4  8  16  32
```

Multiply by 2 to get next entry to right.
Divide by 2 to get next entry to left.

B. Extension of a pattern:

```
      3×¯2 ¯1 0 1 2 3 4
¯6  ¯3  0  3  6  9  12
```

Extension to zero and negative numbers by maintaining pattern (subtraction of 3). This _defines_ multiplication by zero and negative numbers.

```
      2*¯2 ¯1 0 1 2 3 4
0.25  0.5  1  2  4  8  16
```

Extension by division by 2 _defines_ power for zero and negative numbers.

C. Insertion in a pattern:

```
      4*1 2 3 4
4  16  64  256
```

Pattern is multiplication by 4.

```
      4*1 1.5 2 2.5 3 3.5 4
4  8  16  32  64  128  256
```

Pattern is multiplication by a factor such that two applications produces a factor of 4.

```
      2*1 1.5 2 2.5 3
2  2.8284  4  5.6568  8
```

Generalization leads to non-integer results.

Note: The maximum number of digits printed to the right of the decimal point is determined by entering a command of the following form:

```
)DIGITS 4
```

THE POSITIVE INTEGERS

A. The integers to N:

```
      ⍳3
1  2  3

      ⍳5
1  2  3  4  5
```

B. Various functions on the integers:

```
      3+⍳5
4  5  6  7  8
      (⍳5)-3
¯2  ¯1  0  1  2
      +/⍳2
3
      +/⍳3
6
      +/⍳4
10
      2+.1×⍳10
2.1  2.2  2.3  2.4  2.5  2.6  2.7  2.8  2.9  3
```

C. Indexing with the ⍳ function:

```
      P←2 3 5 7 11

      P[⍳3]
2  3  5
      P[2+⍳3]
5  7  11
      P[(2×⍳3)-1]
2  5  11
      ×/P[⍳3]
30
```

D. Repeat some of the work of earlier sections using the iota (⍳) function where possible, e.g., use S←⍳7 instead of S←1 2 3 4 5 6 7.

SUMMATION OF SERIES

A. Enter the following expressions for the summation of series:

```
N←5
S←⍳N

+/S

+/S*2

+/1÷S

+/1÷!0,S
```

B. Repeat Part A for various values of *N* to attempt to determine for each series either what its value is as a simple expression in *N* (for example, +/*S* is equal to .5×*N*×*N*+1) or its limiting value for large values of *N*.

C. The expression -/3 1 4 7 2 is equivalent to 3-(1-(4-(7-2))) and hence is equal to the <u>alternating sum</u> of the list 3 1 4 7 2, that is, to the sum of the numbers in the odd positions less the sum of the numbers in the even positions. More generally, -/*X* is the alternating sum over any list *X*. Pepeat Parts A and B substituting the symbols -/ for each occurrence of the symbols +/.

POWER SERIES

A. Enter the following power series in X:

```
N←5
S←0,⍳N
X←2

+/X*S

-/X*S

+/(÷!S)×X*S

-/(÷!S)×X*S
```

B. Repeat Part A for various values of X and N and attempt to obtain for each a simple equivalent expression in N or an expression for the limiting value for large values of N.

C. Repeat Parts A and B for the following expressions:

```
C←2×S

-/(÷!C)×X*C

-/(÷!C+1)×X*C+1
```

D. The expression 1○X gives the sine of X (for X in radians) and the expression 2○X gives the cosine of X. Use the following expressions to help identify expressions for the limiting values of some of the series of Parts A-C:

```
*X

*-X

1○X

2○X
```

DIFFERENCING A FUNCTION

A. Enter the following expressions:

```
X←0,⍳7
V←X*2

V

1↓V

¯1↓V

(1↓V)-(¯1↓V)
```

B. The expression (1↓V)-(¯1↓V) yields the differences between the successive elements of V which resulted from the square function. A general difference function D can therefore be defined by entering the following:

```
      ∇Z←D V
[1]   Z←(1↓V)-(¯1↓V)∇
```

C. Apply the difference function as follows:

```
D X*2

D X*3

D D X*3

D D D X*3
```

D. Apply the difference function D to any other expressions in X which you may choose.

COMBINATIONS AND BINOMIAL COEFFICIENTS

A. Enter the following expressions:

```
2!3
2!4
2!5
3!5
5!3
7!10
```

B. Construct a table of the dyadic function ! as follows:

```
X←0,⍳7
⌊X∘.!X
```

(The results of *X*∘.!*X* are all integers and therefore the application of the integer part function (⌊) in the foregoing expression does not change the result, but it does cause the table to print more compactly.)

C. Check a number of entries in the table of Part B to verify that the function ! is the "combinations" or "out of" function, that is *A*!*B* yields the number of ways in which *A* things can be chosen from *B*.

D. Evaluate the expression (0,⍳*N*)!*N* for various integer values of *N* and state the significance of the expression.

E. Evaluate the expression +/(0,⍳*N*)!*N* for various values of *N* and write a simple equivalent expression in *N*.

ITERATION

A. Generation of Pascal's Triangle (binomial coefficients):

```
      Z←1
      0,Z
0 1
      Z,0
1 0
      Z←(0,Z)+(Z,0)
      Z
1 1
      Z←(0,Z)+(Z,0)
      Z
1 2 1
      Z←(0,Z)+(Z,0)
1 3 3 1
```

B. A function definition employing iteration:

```
      ∇Z←PA
[1]   Z←1
[2]   Z←(0,Z)+(Z,0)
[3]   Z
[4]   →2∇

      PA
1 1
1 2 1
1 3 3 1            Stop this endless iteration
1 4 6 4 1          by pressing ATTN button.
```

C. An iterative function to produce binomial coefficients:

```
      ∇Z←B N
[1]   Z←1
[2]   Z←(0,Z)+(Z,0)
[3]   →2×N≥ρZ∇

      B 4
1 4 6 4 1
      Q←B 5
      +/Q
32
```

D. Define a function *FIB* such that *FIB N* produces the first *N* Fibonacci numbers. For example:

```
      FIB 8
1 1 2 3 5 8 13 21
```

(Each is the sum of the two preceding it.)

OTHER TOPICS

Topic	Pages	Reference
LOGIC	31- 32	1
SETS	37- 41	1
COORDINATE GEOMETRY	19- 21	1
ITERATION	146-148	2
PRIME NUMBERS	17- 18	1
CALCULUS	26- 27	1
	107-112	3
COMPUTERS	51- 52	1
	58- 63	1
PROGRAMMING	22- 26	4
DRILL	44- 48	4
COMMUTATIVITY	6- 9	5
CLOSURE	9- 10	5
FURTHER NOTATION	Appendix A	6

REFERENCES

1. Iverson, K. E., APL in Exposition, Tech. Report 320-3010, IBM Scientific Center, Philadelphia, 1971.

2. Iverson, K. E., Elementary Algebra, Tech. Report 320-3001, IBM Scientific Center, Philadelphia, 1971.

3. Iverson, K. E., Elementary Functions, Science Research Associates, Chicago, 1966.

4. Iverson, K. E., The Use of APL in Teaching, Publication G320-0996-0, IBM Corp., White Plains, 1969.

5. Berry, P. C., A. D. Falkoff and K. E. Iverson, Using the Computer to Compute: A Direct but Neglected Approach to Teaching Mathematics, Tech. Report 320-2988, IBM Scientific Center, Philadelphia, 1971.

6. Falkoff, A. D. and K. E. Iverson, APL\360-OS and APL\360-DOS User's Manual, Publication SH20-0906-0, IBM Corp., White Plains, 1970.

×	1	2	3	4	5	6	7
1							
2							
3							
4							
5							
6							
7							

+	1	2	3	4	5	6	7
1							
2							
3							
4							
5							
6							
7							

×	1	2	3	4	5	6	7
2							
3							

+	1	2	3	4	5	6	7
2							
3							

=	¯3	¯2	¯1	0	1	2	3
¯3							
¯2							
¯1							
0							
1							
2							
3							

-	1	2	3	4	5	6	7
1							
2							
3							
4							
5							
6							
7							

Make a transparency of this page to use as an overlay on outer product tables to make them more easily readable.

<table>
<tr><th></th><th>NAME</th><th>SYMBOL</th><th>DEFINITION OR EXAMPLE</th><th>SECTION #</th></tr>
<tr><td>D</td><td>Addition</td><td>+</td><td>3+4←→7</td><td>1.2</td></tr>
<tr><td>Y</td><td>Multiplication</td><td>×</td><td>3×4←→12</td><td>1.2</td></tr>
<tr><td>A</td><td>Subtraction</td><td>-</td><td>3-4←→¯1</td><td>3.1</td></tr>
<tr><td>D</td><td>Division</td><td>÷</td><td>3÷4←→.75</td><td>5.1</td></tr>
<tr><td>I</td><td>Maximum</td><td>⌈</td><td>3⌈4←→4</td><td>2.4</td></tr>
<tr><td>C</td><td>Minimum</td><td>⌊</td><td>3⌊4←→3</td><td>2.4</td></tr>
<tr><td></td><td>Power</td><td>⋆</td><td>3⋆4←→81   A⋆B←→×/BρA</td><td>2.5 6.5-6</td></tr>
<tr><td>F</td><td>Remainder</td><td>|</td><td>3|4←→1</td><td>7.1</td></tr>
<tr><td>U</td><td>Relations</td><td><≤=≥>≠</td><td>3<4←→1   4<3←→0</td><td>4.8</td></tr>
<tr><td>N</td><td>Or</td><td>∨</td><td></td><td>14.2</td></tr>
<tr><td>C</td><td>And</td><td>∧</td><td>∨ 0 1 ∧ 0 1 ⍱ 0 1 ⍲ 0 1</td><td>14.2</td></tr>
<tr><td>T</td><td>Not-or</td><td>⍱</td><td>0 0 1 0 0 0 0 1 0 0 1 1</td><td>14.2</td></tr>
<tr><td>I</td><td>Not-and</td><td>⍲</td><td>1 1 1 1 0 1 1 0 0 1 1 0</td><td>14.2</td></tr>
<tr><td>O</td><td>Domino</td><td>⌹</td><td>B⌹M is soln of B=M+.×X</td><td>16.15</td></tr>
<tr><td>N</td><td></td><td></td><td></td><td></td></tr>
<tr><td>S</td><td>Repetition</td><td>ρ</td><td>3ρ5←→5 5 5</td><td>1.7 13.3</td></tr>
<tr><td></td><td>Catenation</td><td>,</td><td>4 2,1 3 5←→4 2 1 3 5</td><td>6.2</td></tr>
<tr><td></td><td>Take</td><td>↑</td><td>2↑4 5 6←→4 5</td><td>10.5</td></tr>
<tr><td></td><td>Drop</td><td>↓</td><td>2↓4 5 6←→6</td><td>10.5</td></tr>
<tr><td></td><td>Compression</td><td>/</td><td>0 1 1 0/1 2 3 4←→2 3</td><td>7.5</td></tr>
<tr><td>M</td><td>Negation</td><td>-</td><td>-4←→¯4</td><td>8.2</td></tr>
<tr><td>O</td><td>Reciprocal</td><td>÷</td><td>÷4←→.25</td><td>8.3</td></tr>
<tr><td>N</td><td>Magnitude</td><td>|</td><td>|¯4←→4</td><td>8.4</td></tr>
<tr><td>A</td><td>Factorial</td><td>!</td><td>!4←→1×2×3×4</td><td>8.1</td></tr>
<tr><td>D</td><td>Ceiling</td><td>⌈</td><td>⌈3.4←→4</td><td>8.5</td></tr>
<tr><td>I</td><td>Floor</td><td>⌊</td><td>⌊3.4←→3</td><td>8.5</td></tr>
<tr><td>C</td><td>Complement</td><td>~</td><td>~1←→0   ~0←→1</td><td>8.6</td></tr>
<tr><td></td><td>Matrix Inverse</td><td>⌹</td><td>M+.×⌹M is the identity</td><td>16.15</td></tr>
<tr><td></td><td>Integers</td><td>⍳</td><td>⍳4←→1 2 3 4</td><td>1.5</td></tr>
<tr><td></td><td>Size</td><td>ρ</td><td>ρ4 1 3 6 2←→5</td><td>8.7</td></tr>
<tr><td></td><td>Flipping</td><td>⌽ ⊖ ⍉</td><td>Flip table about axis</td><td>4.3</td></tr>
<tr><td>O</td><td>Assignment</td><td>←</td><td>X←6</td><td>1.3</td></tr>
<tr><td>T</td><td>Indexing</td><td>X[I]</td><td>2 3 5 7[2 4]←→3 7</td><td>4.4</td></tr>
<tr><td>H</td><td></td><td>M[I;J]</td><td></td><td></td></tr>
<tr><td>E</td><td>Function</td><td>∇Z←F X</td><td></td><td>9.1</td></tr>
<tr><td>R</td><td>Definition</td><td>∇Z←X F Y</td><td></td><td>9.2</td></tr>
<tr><td></td><td>Parentheses</td><td></td><td></td><td>1.2</td></tr>
<tr><td></td><td>Execution order</td><td></td><td>3×4+5-7←→3×(4+(5-7))</td><td>1.2</td></tr>
<tr><td></td><td>Vectors</td><td></td><td>2 3 5×1 2 3←→2 6 15</td><td>1.6</td></tr>
<tr><td></td><td>Tables, Matrices</td><td></td><td></td><td>2.1 13.3</td></tr>
<tr><td></td><td>Reduction (Over)</td><td>f/</td><td>+/2 3 5←→10   ×/3 4←→12</td><td>1.4 4.10</td></tr>
<tr><td></td><td>Outer Product</td><td>∘.f</td><td></td><td>2.3</td></tr>
<tr><td></td><td>Inner Product</td><td>f.g</td><td></td><td>13.2 13.4</td></tr>
</table>

SUMMARY OF NOTATION

Reprinted from Reference 2.

IBM

IBM Cambridge Scientific Center
545 Technology Square
Cambridge, Massachusetts 02139

IBM Houston Scientific Center
6900 Fannin Street
Houston, Texas 77025

IBM Los Angeles Scientific Center
1930 Century Park W.
Los Angeles, California 90067

IBM Palo Alto Scientific Center
2670 Hanover Street
Palo Alto, California 94304

IBM Philadelphia Scientific Center
3401 Market Street
Philadelphia, Pennsylvania 19104